

Doc No.: AN-206

Version: 1.0

Date: 14 April 2008

Subject: Flying shear with variable distance

APPLICATION NOTE

www.triomotion.com

1. Flying Shear

When setting up a linear axis to run a flying shear, higher efficiency can be achieved by making the length of travel vary depending on the requirements of the cutter. If the cutter takes, say, 250 msec to complete a cut cycle, the axis only needs to travel the required distance for this time delay and not the full stroke available. In this way the length of travel will be shorter when the line speed is slow.

The program shown below shows how to use the TrioBASIC command **MOVELINK** to achieve a variable length flying shear operation.

2. Program “FLYSHEAR3”

This program is intended as an example of using the **MOVELINK** command to achieve a flying shear machine for cuts made at various speeds, where the cut cycle is a fixed time. *The **MOVELINK** command specifies a movement on an axis which is position linked to an encoder input.*

Two **MOVELINK** commands are used. The first one specifies the ramp up and the synchronised part of the motion.

```
MOVELINK (ad+sd,ad*2+sd,ad*2,0,conveyor)
```

The second **MOVELINK** specifies only the ramp down.


```
MOVELINK (dd,2*dd,0,2*dd,conveyor)
```

Figure 1 shows how the 2 **MOVELINK**s, in blue and red share the motion. *The controller will automatically bind the two sections of the movement together as if they were one **MOVELINK**.* They are specified separately to allow the first to be cancelled before the end of travel. This allows the overall distance to be shortened while still performing the deceleration phase.

The **CANCEL** command is used to end the first **MOVELINK** at the required time. E.g.


```
MOVELINK (ad+sd,ad*2+sd,ad*2,0,conveyor)  
MOVELINK (dd,2*dd,0,2*dd,conveyor)  
' wait for signal  
CANCEL ' this ends the first movelink and the second one starts.
```

Figure 1.: Motion of the Flying Shear



3. Output control

The most simple way of setting the output to trigger the cutter mechanism in a shear is to use the position of the slide axis which carries the cutter. Provided this axis is homed and the start position is 0, then the position of the start of the synchronised motion can be used as the trigger point for the cutter.



```

MOVELINK(ad+sd,ad*2+sd,ad*2,0,conveyor)
MOVELINK(dd,2*dd,0,2*dd,conveyor)
WAIT UNTIL MPOS>ad+td
OP(8,ON)
' time delay
WA(250)
OP(8,OFF)
CANCEL ' decelerate the slide axis.

```

4. Example program

```

' Example Flying Shear Program:
' Trio Motion Technology. Apl 2008
' This program is designed to allow a flying shear to
' end the move at any point do a controlled deceleration.

' axis "conveyor" is the conveyor axis number
' axis "slide" is the cutter transport axis
slide=0
conv=1

UNITS AXIS(slide)=8000' This is the number of edges/mm etc
UNITS AXIS(conv)=6000

' "ad" is the acceleration distance on the SLIDE
' "sd" is the distance of matched movement
' "dd" is the deceleration distance on the SLIDE
' "td" is the settling distance after the end of the accel phase
ad=100
sd=350
dd=80
td=20

cuts_apart=100' Specify global variable to use for cut length
VR(cuts_apart)=20' Default distance between cuts
IF VR(cuts_apart)<(2*sd+4*ad+4*dd) THEN
  PRINT "Not enough space for moves specified"
  STOP
ENDIF

DEFPOS(0) AXIS(slide)
REP_DIST AXIS(conveyor)=VR(cuts_apart) ' Set the cut repeat distance
REP_OPTION AXIS(conveyor)=1
' NOTE: Program assumes "conveyor" axis is counting UP

BASE(slide)
REPEAT
  ' first MOVELINK starts on position 0 of conveyor
  ' (controls cut length)
  MOVELINK(ad+sd,ad*2+sd,ad*2,0,conveyor,2,0)
  MOVELINK(dd,2*dd,0,2*dd,conveyor) ' decel part
  WAIT UNTIL MPOS>ad+td ' Wait till accel finished + settling distance
  OP(8,ON)' Switch on cutter
  WAIT UNTIL (IN(cutter_up)=ON)
  OP(8,OFF)
  CANCEL
  ' Retract with MOVEABS (set ACCEL and DECEL to suit max speed)
  MOVEABS(0)
UNTIL FALSE

```

```
' Note: This program does not:  
,  
' 1 - Set gains  
' 2 - Datum slide axis  
' 3 - Interact with stop switch etc
```

4.1. Alternative timed cutter operation

```
REPEAT  
  ' first MOVELINK starts on position 0 of conveyor  
  ' (controls cut length)  
  MOVELINK(ad+sd,ad*2+sd,ad*2,0,conveyor,2,0)  
  MOVELINK(dd,2*dd,0,2*dd,conveyor) ' decel part  
  WAIT UNTIL MPOS>ad+td ' Wait till accel finished + settling distance  
  OP(8,ON) ' Switch on cutter  
  WA(cut_time)  
  OP(8,OFF)  
  CANCEL  
  ' Retract with MOVEABS (set ACCEL and DECEL to suit max speed)  
  MOVEABS(0)  
UNTIL FALSE
```

4.2. Alternative timed settling factor

```
REPEAT  
  ' first MOVELINK starts on position 0 of conveyor  
  ' (controls cut length)  
  MOVELINK(ad+sd,ad*2+sd,ad*2,0,conveyor,2,0)  
  MOVELINK(dd,2*dd,0,2*dd,conveyor) ' decel part  
  WAIT UNTIL MPOS>ad ' Wait till accel finished  
  WA(settling_time)  
  OP(8,ON) ' Switch on cutter  
  WAIT UNTIL (IN(cutter_up)=ON)  
  OP(8,OFF)  
  CANCEL  
  ' Retract with MOVEABS (set ACCEL and DECEL to suit max speed)  
  MOVEABS(0)  
UNTIL FALSE
```